

Вещественные числа представляются в виде совокупности мантиссы и порядка в двоичной системе счисления. Обычный размер ячейки — 32 или 64 бита.

Результаты вычислений с вещественными числами приближенные. Переполнение приводит к прерыванию работы процессора.

Вопросы и задания

1. Как в памяти компьютера представляются целые положительные и отрицательные числа?
2. Укажите, каков был бы диапазон значений целых чисел, если бы для их хранения использовалась четырехразрядная ячейка.
3. Запишите внутреннее представление следующих десятичных чисел, используя восьмиразрядную ячейку:
а) 32; б) -32; в) 102; г) -102; д) 126; е) -126.
4. Определите, каким десятичным числам соответствуют следующие двоичные коды восьмиразрядного представления целых чисел.
а) 00010101; б) 11111110; в) 00111111; г) 10101010.

ЕК ЦОР: Часть 2, глава 4, § 17. ЦОР № 1, 2.



§ 20

Что такое электронная таблица

Основные темы параграфа:

- структура электронной таблицы;
- данные в электронной таблице;
- режимы отображения данных.

Теперь снова речь пойдет о таблицах. Это особые таблицы.

Представьте себе, что вы являетесь владельцем небольшого торгового павильона, в котором реализуется молочная продукция. Вам приходится вести самые различные формы учета товара. Пусть, например, один из учетных документов должен выглядеть так, как показано в табл. 4.1.

Таблица 4.1

Учет продажи молочных продуктов

Продукт	Цена	Поставлено	Продано	Осталось	Выручка
Молоко	20	100	100	0	2000
Сметана	10,2	85	70	15	714
Творог	18,5	125	110	15	2035
Йогурт	5,4	250	225	25	1215
Сливки	15,2	50	45	5	684

Обратите внимание на следующую особенность этой таблицы: *в ней есть поля, значения которых вычисляются через значения других полей.* Таким полем является поле «Выручка», значение этого поля равно произведению количества проданного товара на цену, а также поле «Осталось», значение которого вычисляется как разность между количеством поставленного товара и количеством проданного.

Поля: «Продукт», «Цена», «Поставлено», «Продано» являются независимыми. Эти поля содержат **исходные данные** для расчетов.

Ситуация в магазине постоянно меняется: продукты продаются, растет выручка, подвозятся новые партии товара. Если вы хотите постоянно поддерживать в вашем учетном документе достоверную информацию, то придется несколько раз в день вносить в него изменения. При этом вы будете не только изменять исходные данные (поля «Поставлено» и «Продано»), но и пересчитывать вручную значения зависимых полей «Осталось» и «Выручка».

Возможно, что у кого-то из вас появилась такая мысль: вот если бы значения этих полей пересчитывались в таблице автоматически с изменением исходных данных!

Может быть, именно так рассуждали авторы одной из самых замечательных идей в области информационных технологий: идеи **электронной таблицы (ЭТ).**



Прикладные программы, предназначенные для работы с электронными таблицами, называются **табличными процессорами.**

Существует достаточно много разнообразных вариантов табличных процессоров. Однако с точки зрения пользователя они очень похожи друг на друга. Поняв принцип устройства электронной таблицы, легко освоить работу с любым конкретным табличным процессором.

Структура электронной таблицы

Что же представляет собой электронная таблица? Вот как выглядит заполненная электронная таблица с учетным документом (табл. 4.2).

Таблица 4.2

Электронная таблица в режиме отображения формул

	A	B	C	D	E	F
1	Продукт	Цена	Поставлено	Продано	Осталось	Выручка
2	Молоко	20	100	100	=C2-D2	=B2*D2
3	Сметана	10,2	85	70	=C3-D3	=B3*D3
4	Творог	18,5	125	110	=C4-D4	=B4*D4
5	Йогурт	5,4	250	225	=C5-D5	=B5*D5
6	Сливки	15,2	50	45	=C6-D6	=B6*D6

Электронная таблица, подобно шахматной доске, состоит из клеток, которые принято называть **ячейками**. Строки и столбцы таблицы имеют обозначения. Чаще всего строки нумеруются числами, а столбцы обозначаются буквами (буквы латинского алфавита). Как и на шахматной доске, каждая клетка (ячейка) имеет свое **имя (адрес)**, состоящее из имени столбца и номера строки. Например: A1, C13, F24 и т. п.

Но если на шахматной доске всего $8 \times 8 = 64$ клетки, то в электронной таблице ячеек значительно больше. Например, у табличного процессора Microsoft Excel таблица максимального размера содержит 256 столбцов и 65 536 строк. Поскольку в латинском алфавите всего 26 букв, начиная с 27-го столбца, используются двухбуквенные обозначения также в алфавитном порядке:

AA, AB, AC, ..., AZ, BA, BB, BC, ..., BZ, CA, ...

Последний, 256-й столбец имеет имя IV (не путайте с римским числом). Значит, существуют ячейки с такими, например, именами: DL67, HZ10234 и т. п.

Разумеется, столь большая таблица не может вся поместиться на экране. Экран монитора — это окно, через которое пользователь видит только часть таблицы. Но это окно можно переместить в любое ее место.

Данные в электронной таблице

Все данные таблицы размещаются в ячейках. Содержимым ячейки может быть **текст**, **числовое значение** или **формула**. Табличный процессор должен «знать», данное какого типа хранится в конкретной ячейке таблицы, для того чтобы правильно интерпретировать ее содержимое. Текст и числа рассматриваются как константы. Изменить их можно только путем редактирования соответствующих ячеек. Значения же формул автоматически пересчитываются, как только изменится хотя бы один их операнд.

Режимы отображения данных

Таблица 4.2 находится в режиме **отображения формул**, который позволяет проследить алгоритм табличных вычислений. Результаты вычислений по формулам видны на экране в режиме отображения значений. Таблица 4.3 — это та же самая электронная таблица, но переведенная в режим **отображения значений**.

Таблица 4.3

Электронная таблица в режиме отображения значений

	A	B	C	D	E	F
1	Продукт	Цена	Поставлено	Продано	Осталось	Выручка
2	<i>Молоко</i>	20	100	100	0	2000
3	<i>Сметана</i>	10,2	85	70	15	714
4	<i>Творог</i>	18,5	125	110	15	2035
5	<i>Йогурт</i>	5,4	250	225	25	1215
6	<i>Сливки</i>	15,2	50	45	5	684

Коротко о главном

Электронные таблицы предназначены для организации табличных расчетов на компьютере. Прикладные программы, работающие с электронными таблицами, называются табличными процессорами.

Наименьшая структурная единица электронной таблицы — ячейка. Имя ячейки складывается из буквенного имени столбца и номера строки.

В ячейке может помещаться текст (символьная последовательность), число, формула.

Ячейки, в которые пользователь заносит числа, содержат исходные данные для вычислений. В ячейках с формулами получаются результаты вычислений.

Изменение исходных данных мгновенно приводит к пересчету формул, в которые эти данные входят.

Электронные таблицы (так же, как и базы данных) можно рассматривать как информационные модели реальных объектов.

Вопросы и задания

1. Что такое табличный процессор?
2. Как именуются ячейки таблицы? Какая информация может храниться в ячейках?
3. В чем разница между режимом отображения формул и режимом отображения значений?
4. Что происходит в электронной таблице в результате замены числа в ячейке на новое значение?
5. В чем состоит существенное отличие электронной таблицы от таблицы реляционной базы данных? Подготовьте сообщение с примерами.

ЕК ЦОР: Часть 2, глава 4, § 18. ЦОР № 4, 8.



§ 21

Правила заполнения таблицы

Основные темы параграфа:

- *тексты в электронной таблице;*
- *правила записи чисел;*
- *правила записи формул;*
- *подготовка таблицы к расчетам.*

Тексты в электронной таблице

При вводе в ячейку таблицы последовательности символов, которая не может быть воспринята как число или формула, табличный процессор воспринимает ее как текст, т. е. как символьную информацию. Кроме того, любая последовательность, ввод которой начинается с апострофа (‘), воспринимается как текст (апостроф не отображается). В таблицах 4.2 и 4.3 ячейки в первой строке и в столбце А заполнены текстами.

Правила записи чисел

В записях исходных данных, а также в математических формулах присутствуют числа — числовые константы, которые разделяются на **целые** и **вещественные** (действительные). Запись целых числовых констант не вызывает затруднений. Например:

25; -3456; +2134567.

Вещественные константы можно записывать двумя способами: в форме с **фиксированной запятой** (обычная форма) и в форме с **плавающей запятой**.

Запись числовой константы в форме с фиксированной запятой предполагает, что число содержит целую и дробную части, разделенные десятичной *запятой*. Например, числовая константа 3,1415 записывается как 3,1415.



Числовая константа в форме с плавающей запятой трактуется как мантисса, умноженная на 10 в степени, равной порядку.

Например, в записи числа в виде $0,5 \cdot 10^9$ сомножитель 0,5 является мантиссой, а показатель степени 9 является порядком.

При записи в электронную таблицу числовой константы в форме с плавающей запятой сначала пишется **мантисса**, затем — латинская буква E (прописная или строчная), после нее — **порядок**. Мантисса может быть целой константой или константой с фиксированной запятой, а порядок — только целой константой. Порядок указывает, на какое количество позиций и в каком направлении должна сместиться запятая в мантиссе.

Например, математическая запись $0,5 \cdot 10^9$ в электронной таблице выглядит так: 0,5e9; а $1 \cdot 10^{-2}$ запишется как 1e-2.

Обычно форма с плавающей запятой используется для представления очень больших или очень маленьких чисел. Например: 2e+25; 1e-30.

Правила записи формул

Запись формулы в ячейке начинается со знака «равно» (=). Формулы записываются по строго определенным правилам. Эти правила нетрудно освоить. Формулы содержат числа, имена ячеек, знаки операций, круглые скобки, имена функций. Вот как выглядят знаки операций:

- + (сложение);
- (вычитание);
- * (умножение);
- / (деление);
- ^ (возведение в степень).

Вся формула пишется в строку, символы выстраиваются последовательно друг за другом.

Формулы в табл. 4.2 имеют следующий смысл:

C2-D2 — из числа ячейки C2 вычесть число из ячейки D2, результат будет помещен в ячейку E2, в которой записана эта формула;

B2*D2 — число из ячейки B2 умножить на число из ячейки D2, результат будет помещен в ячейку F2.

Вот еще примеры записи формул:

- =2,5*A1 + B2*C3
- =(B3 - C1)/(B3 + C1)
- =F7/2 + G7/3
- =(A5 - 1)^2



Нетрудно понять смысл этих математических выражений. Как всегда, в первую очередь выполняются операции в скобках. При отсутствии скобок последовательность операций определяется их старшинством. По порядку убывания старшинства операции располагаются так:

^ (возведение в степень);

*, / (умножение, деление);

+, - (сложение, вычитание).

Несколько подряд записанных операций одинакового старшинства выполняются в порядке их записи в формуле (слева направо). Например, формула M13/365*N4 будет соответствовать математической записи:

$$\frac{M13}{365} \cdot N4.$$

В формулах допускается употребление некоторых математических функций. Например, математическое выражение

$$\sqrt{B5 + B6}$$

запишется в таком виде:

$$\text{КОРЕНЬ}(B5+B6).$$

Здесь КОРЕНЬ — имя функции «квадратный корень». Аргументы всегда пишутся после имени функции в круглых скобках.

Подготовка таблицы к расчетам

Совсем не обязательно при заполнении электронной таблицы сразу заносить в нее исходные данные. *Таблицу можно предварительно подготовить к вычислениям в виде бланка, не заполненного числами.* Для этого нужно заполнить все ячейки с текстовой информацией и записать в вычисляемые ячейки соответствующие формулы. В режиме отображения значений такая таблица выглядит почти пустой: в вычисляемых ячейках будут высвечиваться нулевые значения. Как только пользователь начнет заносить в нее числовые данные, в зависимых ячейках сразу же будут появляться вычисленные по формулам результаты. В таблице 4.4 приведен пример заготовки рассмотренного учетного документа в режиме отображения формул.

Таблица 4.4

Таблица, подготовленная к расчетам

	A	B	C	D	E	F
1	Продукт	Цена	Поставлено	Продано	Осталось	Выручка
2	Молоко				=C2-D2	=B2*D2
3	Сметана				=C3-D3	=B3*D3
4	Творог				=C4-D4	=B4*D4
5	Йогурт				=C5-D5	=B5*D5
6	Сливки				=C6-D6	=B6*D6

Коротко о главном

При вводе в таблицу любая последовательность символов, которая не может быть числом или формулой, а также вводимая после апострофа, воспринимается как текст.

Числа в электронной таблице представляются в форме с фиксированной запятой и в форме с плавающей запятой.

Формулы могут включать в себя числа, имена ячеек, функции, знаки операций, круглые скобки.

Предварительное занесение в таблицу лишь символьных данных и формул равносильно программированию таблицы для последующих расчетов.

Вопросы и задания

1. Как ввести текст в ячейку электронной таблицы?
2. В каких двух форматах представляются числа? В чем разница между ними?
3. Сформулируйте правила записи формул. Что произойдет, если при вводе формулы вы нарушите эти правила?
4. Как можно заранее подготовить таблицу для вычислений?
5. Запишите в традиционной математической форме следующие формулы из электронной таблицы, предварительно ответив на вопрос, в какой последовательности будут выполняться математические операции.



$$\begin{aligned}
 &=C2+A5/Q3; & &=(C2+A5)/3; & &=C2/(A5+3); \\
 &=A1*A2/D12*DQ3; & &=A1*A2/D12/D3; & &=A1*A2/(D12*D3); \\
 &=B2^2-D3^Q5; & &=F4+(A4*5)^3; & &=F4^3*A4.
 \end{aligned}$$



6. Постройте электронную таблицу «Оплата электроэнергии» для расчета ежемесячной платы за расход электроэнергии в течение года. Исходной информацией являются показания счетчика в начале каждого месяца и стоимость одного киловатт-часа.



ЕК ЦОР: Часть 2, глава 4, § 19. ЦОР № 3, 4.

§ 22

Работа с диапазонами. Относительная адресация

Основные темы параграфа:

- что такое диапазон (блок);
- функции обработки диапазона;
- принцип относительной адресации;
- сортировка таблицы.

Что такое диапазон (блок)

Табличные процессоры позволяют выполнять некоторые вычисления с целой группой ячеек, называемой диапазоном.



Диапазон (блок, фрагмент) — любая прямоугольная часть таблицы.

Обычно диапазон обозначается именами верхней левой и нижней правой ячеек, разделенными двоеточием. Например, в табл. 4.4 диапазон, состоящий из вычисляемых ячеек, обозначается следующим образом: E2:F6 (в табл. 4.4 он выделен тем-

ным фоном). Минимальным диапазоном является одна ячейка таблицы.

Функции обработки диапазона

В каждом табличном процессоре имеется целый набор функций, применяемых к диапазонам. Это **суммирование чисел (СУММ)**, входящих в диапазон, **вычисление среднего значения (СРЗНАЧ)**, **нахождение максимального (МАКС)** и **минимального (МИН) значений** и некоторые другие. Такие функции называются **статистическими**.

Предположим, что в конце рабочего дня необходимо подсчитать выручку, полученную за день от продажи молочных продуктов. Для этого в таблице 4.3 нужно просуммировать все числа из диапазона F2:F6. Пусть функция суммирования обозначается словом СУММ. Тогда нужная нам формула запишется так: =СУММ(F2:F6). Она обозначает следующее:

$$\text{СУММ}(F2:F6)=F2+F3+F4+F5+F6.$$

Запишем формулу суммирования в ячейку F7, а в ячейку E7 — текст «ВСЕГО:». Результат — в табл. 4.5.

Таблица 4.5

Таблица с вычислением суммарной выручки

	A	B	C	D	E	F
1	Продукт	Цена	Поставлено	Продано	Осталось	Выручка
2	Молоко	20	100	100	0	2000
3	Сметана	10,2	85	70	15	714
4	Творог	18,5	125	110	15	2035
5	Йогурт	5,4	250	225	25	1215
6	Сливки	15,2	50	45	5	684
7					ВСЕГО:	6648

Табличные процессоры позволяют манипулировать с диапазонами электронной таблицы. *К операциям манипулирования относятся: удаление, вставка, копирование, перенос, сортировка диапазонов таблицы.* Эти операции выполняются с помощью **команд табличного процессора**. Обычно эти команды пользователь выбирает из **меню команд**.

Принцип относительной адресации

Казалось бы, в результате таких манипуляций расчетные формулы могут стать неверными, поскольку изменятся адреса перемещенных на новое место ячеек. Чтобы такого не происходило, в электронной таблице реализован **принцип относительной адресации**.



Согласно принципу относительной адресации, адреса ячеек, используемые в формулах, определены не абсолютно, а относительно ячейки, в которой располагается формула.

Следствием этого принципа является следующее правило:



Всякое изменение места расположения формулы ведет к автоматическому изменению адресов ячеек в этой формуле.

Поясним сказанное на примере. Пусть при подготовке таблицы для расчета продажи товара на следующий день владелец павильона знает, что в этот день не будут подвозиться сметана и творог. Поэтому две соответствующие строки из табл. 4.4 можно удалить. Это делается с помощью команды вида

УДАЛИТЬ А3:F4

На место удаленных строк сдвигаются строки снизу. В результате таблица преобразуется в табл. 4.6.

Таблица 4.6

Таблица после удаления двух строк

	A	B	C	D	E	F
1	Продукт	Цена	Поставлено	Продано	Осталось	Выручка
2	Молоко				=C2-D2	=B2*D2
3	Йогурт				=C3-D3	=B3*D3
4	Сливки				=C4-D4	=B4*D4

Обратите внимание на две последние строки. В присутствующих в них формулах изменились адреса ячеек. Здесь был учтен сдвиг на две строки вверх; сработал принцип относительной адресации.

Сортировка таблицы

Допустим, владелец торгового павильона хочет узнать, какие товары пользуются наибольшим спросом. Для этого достаточно упорядочить строки таблицы по убыванию чисел в столбце «Продано». Большинство табличных процессоров позволяет производить сортировку (упорядочение) таблицы по какому-либо признаку. Для нашего примера формируется команда такого вида:

СОРТИРОВАТЬ СТОЛБЕЦ D ПО УБЫВАНИЮ

Применение этой команды к табл. 4.5 в режиме отображения значений даст результат, показанный в табл. 4.7.

Отсюда видно, что наибольшим спросом пользуется йогурт, а меньше всего покупают сливки.

Эта же таблица в режиме отображения формул — табл. 4.8.

Таблица 4.7

Результат сортировки таблицы по столбцу «Продано»

	A	B	C	D	E	F
1	Продукт	Цена	Поставлено	Продано	Осталось	Выручка
2	<i>Йогурт</i>	5,4	250	225	25	1215
3	<i>Творог</i>	18,5	125	110	15	2035
4	<i>Молоко</i>	20	100	100	0	2000
5	<i>Сметана</i>	10,2	85	70	15	714
6	<i>Сливки</i>	15,2	50	45	5	684
7					ВСЕГО:	6648

Таблица 4.8

Отсортированная таблица в режиме отображения формул

	A	B	C	D	E	F
1	Продукт	Цена	Поставлено	Продано	Осталось	Выручка
2	<i>Йогурт</i>	5,4	250	225	=C2-D2	=B2*D2
3	<i>Творог</i>	18,5	125	110	=C3-D3	=B3*D3
4	<i>Молоко</i>	20	100	100	=C4-D4	=B4*D4
5	<i>Сметана</i>	10,2	85	70	=C5-D5	=B5*D5
6	<i>Сливки</i>	15,2	50	45	=C6-D6	=B6*D6
7					ВСЕГО:	=СУММ(F2:F6)

Снова сработал принцип относительной адресации. Формулы изменились в соответствии с изменением места расположения строк.

Коротко о главном

Диапазон (блок, фрагмент) таблицы — это любая ее прямоугольная часть (в том числе часть строки, часть столбца или одна ячейка).

Для выполнения расчетов над диапазонами, содержащими числа, используются статистические функции: суммирование, усреднение, нахождение наибольшего и наименьшего значений и др.

С диапазонами можно производить операции манипулирования: удаление, вставку, перенос, сортировку.

Принцип относительной адресации: адреса ячеек в формуле определены не абсолютно, а относительно места нахождения этой формулы. Следствие: при перемещении формулы в другую ячейку соответствующим образом изменяются адреса ячеек, содержащиеся в ней.

**Вопросы и задания**

1. Что такое диапазон? Как он обозначается?
2. Какие вычисления можно выполнять над целым диапазоном?
3. Что понимается под манипулированием диапазонами ЭТ?
4. Что такое принцип относительной адресации? В каких ситуациях он проявляется?

5. В ячейке D7 записана формула $(C3+C5)/D6$. Как она изменится при переносе этой формулы в ячейку:
а) D8; б) E7; в) C6; г) F10?
6. В ячейке E4 находится формула СУММ(A4:D4). Куда она переместится и как изменится при:
а) удалении строки 2; б) удалении строки 7; в) вставке пустой строки перед строкой 4; г) удалении столбца C; д) вставке пустого столбца перед столбцом F?
7. К таблице «Оплата электроэнергии», полученной при выполнении задания 6 из предыдущего параграфа, добавьте расчет всей выплаченной за год суммы денег и сумм, выплаченных за каждый квартал (квартал — 3 месяца).

ЕК ЦОР: Часть 2, глава 4, § 20. ЦОР № 2, 3, 8.

§ 23

Деловая графика. Условная функция

Основные темы параграфа:

- *графические возможности табличного процессора;*
- *типы диаграмм;*
- *условная функция.*

Графические возможности табличного процессора

Замечательным свойством электронных таблиц является возможность графического представления числовой информации, содержащейся в таблице. Для этого существует специальный **графический режим** работы табличного процессора. Графики придают наглядность числовым зависимостям.

Типы диаграмм

Табличные процессоры дают возможность получать самые различные формы диаграмм и графиков. Ниже на рисунках показаны два типа диаграмм: **круговая** на рис. 4.1 и **столбчатая** на рис. 4.2. Исходные данные для этих диаграмм извлекаются из одинаковых диапазонов ячеек A2:A6 и D2:D6 таблиц из предыдущего параграфа. Первый диапазон содержит названия продуктов, второй — количество проданных единиц каждого продукта. Из диаграмм сразу видно, что наибольшим спросом у покупателей пользуется йогурт.




Рис. 4.1. Круговая диаграмма

Круговую диаграмму обычно используют в тех случаях, когда нужно показать, какую часть от целого (круга) составляют отдельные величины (секторы). Столбчатая диаграмма (гистограмма) позволяет наглядно сопоставить между собой отдельные величины.




Рис. 4.2. Столбчатая диаграмма (гистограмма)

Условная функция

Продолжим обсуждение задачи об учете продажи молочных продуктов в торговом павильоне. В случае если тот или иной продукт продан полностью, необходимо организовать его подвоз в торговый павильон. Чтобы отразить это в электронной таблице, добавим в табл. 4.8 новый столбец с названием «Подвоз». В ячейках этого столбца будет высвечиваться слово «Да», если подвоз соответствующего продукта необходим, и «Нет», если продукт подвозить не надо. Разумеется, значения «Да» или «Нет» табличный процессор должен определить сам автоматически.

Для решения задачи воспользуемся **условной функцией**. Общий вид условной функции следующий:

ЕСЛИ(<условие>; <выражение 1>; <выражение 2>)

<Условие> — это логическое выражение, которое может принимать значение **ИСТИНА** или **ЛОЖЬ**. С логическими выражениями вы познакомились в главе о базах данных. В электронных таблицах они имеют тот же смысл. <Выражение 1> и <выражение 2> могут быть числами, формулами или текстами.



Условная функция, записанная в ячейку таблицы, выполняется так: если <условие> истинно, то значение данной ячейки определяет <выражение 1>, в противном случае — <выражение 2>.

В нашем случае условие означает проверку на равенство нулю количества оставшегося продукта. В качестве выражений 1 и 2 выступают текстовые константы «Да» и «Нет».

После внесенных изменений учетный документ примет вид, представленный в табл. 4.9 (в режиме отображения формул) и в табл. 4.10 (в режиме отображения значений).

Коротко о главном

В современных табличных процессорах реализована деловая графика: возможность построения диаграмм и графиков по числовым данным в таблице.

Условная функция имеет следующий формат:

ЕСЛИ(<условие>; <выражение 1>; <выражение 2>)

Здесь <условие> — логическое выражение. Если <условие> истинно, то значение ячейки определяет <выражение 1>, если ложно — <выражение 2>.

Вопросы и задания

1. Что такое деловая графика?
2. Какой вид имеет условная функция? Как она выполняется?
3. Продолжите выполнение проектного задания. На основании таблицы «Оплата электроэнергии» (задание 6 из § 21) постройте столбчатую диаграмму, отражающую ежемесячный расход электроэнергии в течение года.

По таблице «Оплата электроэнергии» с добавленным расчетом выплаченной суммы денег за каждый квартал (задание 7 из § 22) получите круговую диаграмму, отражающую относительные размеры выплат в каждом квартале.

К таблице «Оплата электроэнергии» добавьте расчет среднемесячной платы, а также придумайте и реализуйте способ подсчета количества месяцев, плата за которые была выше среднемесячной.

ЕК ЦОР: Часть 2, глава 4, § 21. ЦОР № 8.



Таблица 4.9
Таблица с условной функцией в режиме отображения формул

A	B	C	D	E	F	G
1	Продукт	Цена	Продано	Осталось	Выручка	Подвоз
2	Йогурт	5,4	250	=C2-D2	=B2*D2	=ЕСЛИ(E2=0;"Да";"Нет")
3	Творог	18,5	125	=C3-D3	=B3*D3	=ЕСЛИ(E3=0;"Да";"Нет")
4	Молоко	20	100	=C4-D4	=B4*D4	=ЕСЛИ(E4=0;"Да";"Нет")
5	Сметана	10,2	85	=C5-D5	=B5*D5	=ЕСЛИ(E5=0;"Да";"Нет")
6	Сливки	15,2	50	=C6-D6	=B6*D6	=ЕСЛИ(E6=0;"Да";"Нет")
7				ВСЕГО:	=СУММ(F2:F6)	

Таблица 4.10

Таблица с условной функцией в режиме отображения значений

A	B	C	D	E	F	G	
1	Продукт	Цена	Поставлено	Продано	Осталось	Выручка	Подвоз
2	Йогурт	5,4	250	225	25	1215	Нет
3	Творог	18,5	125	110	15	2085	Нет
4	Молоко	20	100	100	0	2000	Да
5	Сметана	10,2	85	70	15	714	Нет
6	Сливки	15,2	50	45	5	684	Нет
7				ВСЕГО:		6648	

§ 24

Логические функции и абсолютные адреса

Основные темы параграфа:

- запись и выполнение логических функций;
- абсолютные адреса;
- функция времени.

Запись и выполнение логических функций

Продолжим совершенствование таблицы учета продажи молочных продуктов. В условии подвоза товара желательно учесть следующее обстоятельство: подвозить товар не имеет смысла, если торговый павильон заканчивает работу. Это тоже можно предусмотреть в электронной таблице. Ячейку E9 будем использовать для хранения значения времени (в часах), оставшегося до конца рабочего дня. Условие подвоза товара сформулируем так: *товар подвозить, если оставшееся его количество равно нулю И до конца рабочего дня осталось больше двух часов.*

При записи сформулированного выше условия в форме логического выражения должна быть использована логическая операция И (конъюнкция, логическое умножение). Работая с базами данных, вы познакомились с логическими операциями. Однако в электронных таблицах несколько иные правила записи логических выражений, содержащих логические операции.



В электронных таблицах логические операции (И, ИЛИ, НЕ) рассматриваются как логические функции.

Например, логическое выражение, которое принимает значение ИСТИНА, если выполняется сформулированное выше условие подвоза товара, пишется следующим образом (для второй строки, т. е. для йогурта):

$$\text{И}(E2=0; E9>2)$$

Перед скобками ставится имя логической операции (функции), а в скобках — логические операнды.

Следовательно, теперь условная функция в ячейке G2 должна выглядеть так:

$$\text{ЕСЛИ}(\text{И}(E2=0; E9>2); \text{"Да"}; \text{"Нет"})$$

Но в этой формуле таится опасность. Вам уже известно, что при любых манипуляциях с таблицей, связанных с переносом формул в другие ячейки, происходит изменение адресов переменных. Работает принцип относительной адресации. Однако в данном случае адрес ячейки E9 не должен изменяться в формуле. Иначе говоря, этот адрес должен быть не относительным, а **абсолютным**.

Абсолютные адреса



Существует способ «замораживания» адресов в электронных таблицах. На «замороженный» в формуле адрес ячейки не распространяется принцип относительности. Обычно для этой цели используется значок «\$». Можно «заморозить» только номер строки или только имя столбца. Чтобы адрес ячейки сделать абсолютным (неизменным при любом переносе формулы в таблице), нужно знак «\$» писать дважды: \$E\$9.

Теперь должно быть понятно, что условную функцию, решающую вопрос о подвозе товара, следует записать так:

ЕСЛИ(И(E2=0;\$E\$9>2);"Да";"Нет")

Функция времени

Осталось обсудить формулу, вычисляющую количество времени, оставшееся до конца рабочего дня. Можно, посмотрев на часы, вручную вставить это время в ячейку E9. Но в современных табличных процессорах существуют специальные функции (функции времени), позволяющие получить текущее время. Это возможно благодаря тому, что в состав аппаратной части компьютеров входит **таймер** — внутренние часы компьютера. Например, если рабочий день заканчивается в 20 часов, то формула должна быть такой: 20 – ТЕКУЩИЙ ЧАС. Пусть, например, функция определения текущего часа записывается так: ЧАС(ТДАТА()).

После внесения изменений таблица в режиме отображения формул примет вид табл. 4.11, а в режиме отображения значений — табл. 4.12.

Обратите внимание на то, что из табл. 4.12 следует, что молоко уже закончилось, но подвозить его не следует, так как до конца рабочего дня осталось 2 часа.

Таблица 4.11

Таблица в режиме отображения формул

A	B	C	D	E	F	G
1	Продукт	Поставлено	Продано	Осталось	Выручка	Подвоз
2	Йогурт	5,4	225	=C2-D2	=B2*D2	=ЕСЛИ(И(E2=0;\$E\$9>2);"Да";"Нет")
3	Творог	18,5	110	=C3-D3	=B3*D3	=ЕСЛИ(И(E3=0;\$E\$9>2);"Да";"Нет")
4	Молоко	20,0	100	=C4-D4	=B4*D4	=ЕСЛИ(И(E4=0;\$E\$9>2);"Да";"Нет")
5	Сметана	10,2	70	=C5-D5	=B5*D5	=ЕСЛИ(И(E5=0;\$E\$9>2);"Да";"Нет")
6	Сливки	15,2	45	=C6-D6	=B6*D6	=ЕСЛИ(И(E6=0;\$E\$9>2);"Да";"Нет")
7				ВСЕГО:	=СУММ(F2:F6)	
8						
9	Осталось	до конца	дня:	=20-ЧАС(ТДАТА())	часа	

Таблица 4.12

Таблица в режиме отображения значений

A	B	C	D	E	F	G	
1	Продукт	Цена	Поставлено	Продано	Осталось	Выручка	Подвоз
2	Йогурт	5,4	250	225	25	1215	Нет
3	Творог	18,5	125	110	15	2035	Нет
4	Молоко	20,0	100	100	0	2000	Нет
5	Сметана	10,2	85	70	15	714	Нет
6	Сливки	15,2	50	45	5	684	Нет
7				ВСЕГО:		6648	
8							
9	Осталось	до конца	дня:	2	часа		

Коротко о главном

При записи логических выражений можно пользоваться логическими операциями: И, ИЛИ, НЕ. В электронных таблицах логические операции употребляются как функции.

Возможно «замораживание» адресов ячеек, используемых в формулах. «Замороженный» адрес становится абсолютным, т. е. на него не распространяется принцип относительной адресации.



Вопросы и задания

1. Как в электронной таблице реализуются логические операции при записи условных функций?
2. Что такое абсолютный адрес?
3. Продолжите выполнение проектного задания. В таблице «Оплата электроэнергии» (задание 6 из § 21) используйте абсолютный адрес для ячейки, хранящей стоимость 1 кВт·ч электроэнергии. В таблице «Оплата электроэнергии» используйте следующее правило для подсчета суммы оплаты: если израсходовано не более 100 кВт·ч, то цена 1 кВт·ч равна 1,5 руб.; если израсходовано более 100, но менее 300 кВт·ч, то цена — 1,8 руб.; если израсходовано не менее 300 кВт·ч, то цена 1 кВт·ч равна 2 руб. Используйте логические функции.



ЕК ЦОР: Часть 2, глава 4, § 22. ЦОР № 2, 3, 9.

§ 25

Электронные таблицы и математическое моделирование

Основные темы параграфа:

- математическое моделирование;
- этапы математического моделирования на компьютере;
- пример математического моделирования в электронных таблицах.

Математическое моделирование

Что такое компьютерное математическое моделирование? Снова вернемся к теме математического моделирования, обсуждение которой было начато в § 9. Реальную систему, для которой создается математическая модель, принято называть объектом моделирования. Объектами математического моделирования могут быть некоторые конструкции, например железнодорожный мост или корабль; природные объекты, например месторождение полезных ископаемых, водохранилище, а также процессы и явления, происходящие во времени, например взлет космической ракеты с космодрома, изменение погодных условий в определенной географической точке, изменение со временем численности определенных популяций.

Для людей могут оказаться жизненно важными многие вопросы, связанные с этими объектами и процессами. Например: на какой высоте ракета достигнет первой космической скорости и выйдет на орбиту спутника Земли; до какой предельной температуры нагреется ее оболочка? Какой может быть максимальная нагрузка на железнодорожный мост, при которой не будет происходить его разрушение? Каким будет уровень воды в водохранилище в тех погодных условиях, которые предсказывают метеорологи? Не вымрет ли данная популяция животных через сто лет?

На эти вопросы желательно получить ответы теоретическим путем, поскольку экспериментальный путь либо невозможен, либо возможен, но опасен. Например, при перегрузке моста можно его разрушить, при перегреве корпуса ракеты ее можно сжечь; а экспериментально проверить, что будет с популяцией животных через сто лет, невозможно. В подобных ситуациях на помощь человеку приходят математическое моделирование и вычислительный эксперимент.

Этапы математического моделирования на компьютере

В математической модели используются количественные (числовые) характеристики объекта. Например, в математической модели полета ракеты учитываются масса и скорость ракеты, сила тяги двигателей, сопротивление атмосферного воздуха, теплоемкость обшивки ракеты, время полета, высота ракеты над поверхностью Земли, плотность атмосферы. Все эти величины связываются между собой через уравнения, отражающие физические законы движения тела в воздушной среде, нагревания



тела в процессе трения. Из этих уравнений, зная одни величины — исходные данные, можно вычислить другие величины — результаты. Например, зная массу ракеты, силу тяги двигателей, скорость сгорания топлива, коэффициент трения воздуха о корпус, можно вычислить, какой будет высота и скорость ракеты в данный момент времени, а также температура обшивки ракеты. Часто такие расчеты бывает трудно осуществить вручную, и тогда используются компьютерные методы решения задачи.

Повторим определения понятий, которые были введены в § 9.



Реализованная на компьютере математическая модель называется **компьютерной математической моделью**, а проведение расчетов с помощью компьютерной модели с целью прогнозирования поведения моделируемой системы называется **вычислительным экспериментом**.

Таким образом, этапы компьютерного математического моделирования следующие:

- 1) выделение количественных характеристик моделируемой системы, существенных для решаемой задачи;
- 2) получение математических соотношений (формул, уравнений, систем уравнений и пр.), связывающих эти характеристики;
- 3) определение способа решения полученной математической задачи и реализация ее на компьютере с помощью прикладных программных средств или на языках программирования;
- 4) решение поставленной задачи путем проведения вычислительного эксперимента.

В результате вычислительного эксперимента можно получить прогноз поведения исследуемой системы; выяснить, как изменение одних характеристик системы отразится на других.

Одним из видов прикладных программных средств, пригодных для реализации математической модели на компьютере, являются табличные процессоры.

Пример математического моделирования в электронных таблицах

Чаще всего электронные таблицы используются в задачах такого типа, которые были рассмотрены в предыдущих параграфах: для получения расчетных ведомостей, смет, справок, списков, т. е. в области делопроизводства. Однако электронные таблицы могут оказаться полезными и для научных целей. С их помощью можно строить компьютерные математические модели, проводить вычислительные эксперименты. Рассмотрим пример такого вычислительного эксперимента.

Ученые установили, что прирост какого-либо вида живых организмов за счет рождаемости прямо пропорционален их количеству, а убыль за счет смертности прямо пропорциональна квадрату их количества. Этот закон известен под названием *закона Мальтуса*.

Пусть в одном хозяйстве собираются разводить карпов. Прежде чем запускать мальков в пруд, решили провести расчеты. Согласно закону Мальтуса, изменение числа рыб за один год вычисляется по формуле

$$\Delta N = kN - qN^2.$$

Здесь N — число карпов в начале года, k — коэффициент прироста, q — коэффициент смертности. Экспериментально установлено, что для данного вида рыб (карпы) и в данных условиях (состояние водоема, наличие корма) $k = 1$, $q = 0,001$.

Если первоначально в пруд запущено N_0 рыб, то из закона следует, что количество карпов через год будет таким:

$$N_1 = N_0 + (kN_0 - qN_0^2).$$

Через два года:

$$N_2 = N_1 + (kN_1 - qN_1^2)$$

и т. д. Можно написать общую формулу для вычисления количества рыб в i -м году после их запуска:

$$N_i = N_{i-1} + (kN_{i-1} - qN_{i-1}^2) \quad \text{для } i = 1, 2, 3, \dots$$

Эта формула является математической моделью процесса размножения рыб в водоеме.



Заполним электронную таблицу для проведения по этой формуле расчета рыбного «поголовья» в пруду в течение нескольких лет — табл. 4.13.

Не надо думать, что всю таблицу приходится вводить посимвольно с клавиатуры. Строки, начиная с 7-й, формируются путем копирования предыдущей строки. При этом относительные адреса изменяются автоматически.

Для получения результатов достаточно занести в ячейку F1 первоначальное число рыб.

Таблица 4.13

Расчет числа рыб в пруду с интервалом в год

	A	B	C	D	E	F
1	$k =$	1	$q =$	0,001	$N =$	
2						
3	Год		Число рыб			
4						
5	1		$=F1 + \$B\$1 * F1 - \$D\$1 * F1 * F1$			
6	$=A5+1$		$=C5 + \$B\$1 * C5 - \$D\$1 * C5 * C5$			
7	$=A6+1$		$=C6 + \$B\$1 * C6 - \$D\$1 * C6 * C6$			
8	$=A7+1$		$=C7 + \$B\$1 * C7 - \$D\$1 * C7 * C7$			
9	$=A8+1$		$=C8 + \$B\$1 * C8 - \$D\$1 * C8 * C8$			
...			

Теперь можно экспериментировать. Проследим, как за 10 лет будет меняться число карпов при разном количестве первоначально запущенных рыб. Вот несколько таблиц с результатами таких расчетов.

$k=1$ $q=0,001$ $N=100$	
Год	Число рыб
1	190
2	343
3	569
4	814
5	965
6	998
7	1000
8	1000
9	1000
10	1000

$k=1$ $q=0,001$ $N=1000$	
Год	Число рыб
1	1000
2	1000
3	1000
4	1000
5	1000
6	1000
7	1000
8	1000
9	1000
10	1000

$k=1$ $q=0,001$ $N=1500$	
Год	Число рыб
1	750
2	937
3	996
4	1000
5	1000
6	1000
7	1000
8	1000
9	1000
10	1000

$k=1$ $q=0,001$ $N=2000$	
Год	Число рыб
1	0
2	0
3	0
4	0
5	0
6	0
7	0
8	0
9	0
10	0

Не правда ли, удивительные результаты? Из приведенных таблиц следует, что невозможно иметь в пруду 2000 карпов и более. Если начальное число рыб меньше 1000, то оно постепенно будет расти до 1000 штук и далее меняться не будет. Если сразу запустить 1000 рыб, то это количество останется неизменным и в последующие годы. Даже если запустить сначала 1500 рыб, то через год их численность сократится в два раза, а затем все равно дойдет до 1000. Если же запустить в пруд 2000 рыб, то через год все они вымрут.

Из полученных результатов рыбоводы могут сделать практические выводы. Приведенные выше таблицы автоматически получались после изменений значения всего лишь в одной ячейке F1.

Коротко о главном

Математической моделью называется информационная модель объекта, выраженная математическими средствами (формулами, уравнениями и т. п.).

Табличный процессор может применяться в качестве инструмента для математического моделирования.

Полученную математическую модель можно использовать для проведения вычислительного эксперимента. Вычислительный эксперимент — это расчеты с помощью компьютерной математической модели с целью прогноза поведения какой-то системы, с целью выяснения вопроса о том, как изменение одних характеристик системы отражается на других.



Вопросы и задания

1. Что такое математическая модель?
2. Что такое вычислительный эксперимент?
3. Проведите вычислительный эксперимент в таблице расчета количества рыб в пруду, поставив следующую цель: подобрать такие значения параметров k и q , при которых количество рыб за 10 лет может быть доведено до 2000.

К решению задачи добавьте графическую обработку результатов: график изменения численности рыб с течением времени.

§ 26

Пример имитационной модели

Основные темы параграфа:

- что такое имитационная модель;
- пример имитационного моделирования в электронной таблице.

Что такое имитационная модель

В предыдущем параграфе вы познакомились с примером реализации в электронных таблицах математической модели. Сейчас рассмотрим пример реализации другого типа модели, которая называется **имитационной моделью**.

Об имитационных моделях говорилось в § 9. Вспомним данное там определение:



Имитационная модель воспроизводит поведение сложной системы, элементы которой могут вести себя случайным образом, иначе говоря, поведение которых заранее предсказать нельзя.

**Пример имитационного моделирования в электронной таблице**

Как и в предыдущем параграфе, пример возьмем из класса моделей, описывающих эволюцию популяций.

Пусть на определенном пространстве случайным образом расселяются живые организмы. В дальнейшем происходит процесс смены поколений: в каких-то местах расселения жизнь сохраняется, в каких-то исчезает. Эти процессы протекают в соответствии с законами эволюции. Законы эволюции в описании модели представляются в виде формальных правил. *Цель моделирования* — проследить изменения в расселении живых организмов со сменой поколений.

Сначала рассмотрим простейший вариант задачи: жизненное пространство одномерное. Это значит, живые организмы расселяются вдоль линии. Будем считать жизненное пространство ограниченным, т. е. рассмотрим отрезок. Отрезок разделяется на ячейки, в пределах каждой из которых может поселиться один организм. Договоримся, что самые крайние ячейки не заселяются. Они определяют границу жизненного пространства.



На рисунке 4.3 показано первоначальное расселение организмов на поле, состоящем из 20 ячеек. Организмы поселились в ячейках с номерами 5, 8 и 12. Ячейки 1 и 20 всегда должны быть пустыми.




Рис. 4.3. Первоначальное расселение организмов

Теперь сформулируем законы эволюции. В следующем поколении в пустой ячейке жизнь может либо появиться, либо нет. В заселенной ячейке жизнь может либо сохраниться, либо исчезнуть. На состояние данной ячейки влияют ее ближайшие соседи: два соседа слева и два соседа справа. Если ячейка была заселена и число живых соседей не превышает двух, то в следующем поколении в этой ячейке жизнь сохранится, иначе жизнь исчезнет (погибнет от перенаселения). Если в ячейке жизни не было, но среди ее соседей есть 1, 2 или 3 живые ячейки, то в следующем поколении в этой ячейке появится жизнь. В противном случае ячейка останется пустой.

Следует учитывать, что у ячеек, расположенных у края, число соседей меньше других. У ячейки номер 2 соседи: 1, 3 и 4. Но ячейка 1 всегда пустая. У ячейки номер 3 из четырех соседей живыми могут быть не больше трех (2, 4, 5). Аналогичная ситуация у крайних правых ячеек.

То, что сказано выше, есть модельное описание процесса эволюции популяции. Формализуем это описание. Распределение живых организмов по ячейкам будем кодировать последовательностью из нулей и единиц. Ноль обозначает пустую ячейку, единица — живую. Например, расселение, отображенное на рис. 4.3, кодируется следующим образом:

00001001000100000000

Номер ячейки обозначим n , а двоичное число, соответствующее этой ячейке в текущем поколении, обозначим $R(n)$. В рассматриваемом примере $R(5) = R(8) = R(12) = 1$. Все остальные значения ячеек равны нулю.

Значения кода в n -й ячейке для следующего поколения будем обозначать $S(n)$. Внимательно проанализировав сформулированные выше правила эволюции, приходим к следующей формуле:

$$\text{Если } 1 \leq R(n-2) + R(n-1) + R(n) + R(n+1) + R(n+2) \leq 3, \\ \text{то } S(n) = 1, \text{ иначе } S(n) = 0.$$

Эта формула работает для значений n от 3 до 18. Всегда: $S(1) = S(20) = 0$. Для ячеек с номерами 2 и 19 в данной сумме нужно убрать по одному слагаемому. Но можно поступить иначе, чтобы оставить справедливой данную формулу для всех ячеек жизненного пространства. Для этого к отрезку добавим по одной фиктивной ячейке справа и слева. Их номера будут, соответственно, 0 и 21. В этих ячейках, как и в ячейках 1 и 20, всегда будут храниться нули. Тогда написанную формулу можно применять для n от 2 до 19.

Итак, модель построена и формализована. Однако имитационной моделью она станет только в результате реализации с помощью какого-то программного компьютерного средства. В качестве такого средства выберем табличный процессор.

Моделью жизненного пространства будет строка электронной таблицы. Первая строка — первое поколение, вторая строка — второе поколение и т. д. Тогда номера ячеек будут идентифицироваться именами столбцов таблицы. Ячейка 0 — столбец А, ячейка 1 — столбец В и т. д., ячейка 21 — столбец V.

В первой строке выставим единицы в ячейках, заселенных в первом поколении. Это будут ячейки F1, I1, M1. Значения незаполненных ячеек по умолчанию приравниваются к нулю.

Теперь в ячейки второй строки нужно записать формулы. Сделать это достаточно один раз. Например, в ячейку C2 занести следующую формулу

$$=ЕСЛИ(И(А1+В1+С1+D1+E1>=1; \\ А1+В1+С1+D1+E1<=3);1;0)$$

Далее, скопировав эту формулу во все остальные ячейки второй строки с D2 по T2, получим картину распределения живых организмов во втором поколении.

Чтобы получить третье поколение, достаточно скопировать вторую строку (блок C2:T2) в третью строку (блок C3:T3). Так можно продолжать сколько угодно.

На рисунке 4.4 показаны результаты имитационного моделирования процесса эволюции исходного расселения живых организмов вплоть до 10-го поколения. Все очень наглядно. Обратите внимание, как драматично развивались события!

В шестом поколении наступило состояние перенаселения. В результате в седьмом поколении вымерли все организмы, кроме крайних. Их спасло свободное пространство слева и справа. От них пошла новая волна жизни!




	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T	U	V
1						1			1				1									
2			0	1	1	1	1	1	1	1	1	1	1	1	1	0	0	0	0	0		
3			1	1	0	0	0	0	0	0	0	0	0	0	1	1	1	0	0	0		
4			1	1	1	1	0	0	0	0	0	0	1	1	1	1	1	1	1	0		
5			1	0	0	1	1	1	0	0	1	1	1	0	0	0	0	0	0	1	1	
6			1	1	1	1	1	1	1	1	1	1	1	1	1	0	1	1	1	1		
7			1	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	1	
8			1	1	1	0	0	0	0	0	0	0	0	0	0	0	0	1	1	1		
9			1	1	1	1	1	0	0	0	0	0	0	0	0	1	1	1	1	1		
10			1	0	0	0	1	1	1	0	0	0	0	1	1	1	0	0	0	1		

Рис. 4.4. Имитация в электронной таблице эволюции десяти поколений популяции живых организмов

Рассмотренная задача является упрощенным (одномерным) вариантом известной модели Дж. Конуэя, которая называется «Жизнь». В этой модели эволюция популяции живых организмов происходит в двумерном пространстве. Рассматривается прямоугольная область, разделенная на квадратные ячейки. Тогда у каждой внутренней ячейки имеются 8 соседей. Судьба жизни в ячейке также зависит от состояния соседних клеток. Но теперь правила эволюции такие: если клетка живая и в ее окружении более трех живых клеток, то она погибает от перенаселения; если же живых соседей меньше двух, то она погибает от одиночества. В пустой клетке в следующем поколении зарождается жизнь, если у нее есть ровно три живых соседа.

Попробуйте самостоятельно получить имитационную модель для этой задачи в среде электронной таблицы. Последовательность действий будет аналогичной рассмотренной в примере. По-прежнему для перехода к новому поколению нужно использовать метод копирования диапазона. Но только теперь придется копировать не линейный диапазон, а прямоугольный.



Коротко о главном

Рассмотренная имитационная модель эволюционного типа позволяет проследить за изменениями в расселении живых организмов со сменой поколений.

Удобство применения электронной таблицы для имитационного моделирования заключается в простоте реализации вычислительного алгоритма и наглядности представления результатов.

Вопросы и задания

1. Где используется метод имитационного моделирования?
2. В чем отличие эволюционной задачи, решавшейся методом математического моделирования в § 25, от задачи, решавшейся в данном параграфе методом имитационного моделирования?
3. Проведите вычислительный эксперимент на линейной имитационной модели для различных вариантов исходного расселения организмов, отвечающий на следующие вопросы.
 - а) Возможно ли такое расселение, при котором все организмы в конце концов вымрут?
 - б) Не ведет ли любое расселение в конечном итоге к одной и той же последовательности поколений?
 - в) Что меняется с изменением размера жизненного пространства?
- 4¹⁾. Постройте в электронных таблицах двумерную модель «Жизнь». Проведите вычислительный эксперимент с разными вариантами первоначального расселения организмов.

Попробуйте найти такие первоначальные расселения, которые:

 - а) обречены на гибель;
 - б) не меняются со сменой поколений;
 - в) ведут к периодической смене повторяющихся конфигураций расселения.

ЕК ЦОР: Часть 2, глава 4, § 24. ЦОР № 1, 3.



¹⁾ Задание повышенной сложности.